1

Supplementary Data

S2. Model Code

PROGRAM: mnfull.csl 

! Mnfull.csl PBPK model with tissue binding and brain preferential influx

! 

! The model inputs concentrations in ug/ml, diet in ppm and inhalation

! in mg/m3. It converts all inputs to ug or ug/hr for the calculations.

! After integrating to account for time dependent changes, the tissue concentrations

! are calculated as ug/ml for comparisons with data sets. Tracer kinetics simulate

! whole body Mn54 (kBq) based on exposure in Ci, assuming that tracer mass is negligable.

!

! Dose dependant induced biliary and inlfux rate constants are added to capture 

! changes in controls during high inhaled dose

!

! Multiple routes of exposures for tracer kinetics (oral, iv, ip, inhaled and

! subcutaneous) to reproduce the simulations for Dastur et al. and Furchner et al. and Newland et al.

!

INITIAL

!************************************

! Physiological parameters

!************************************

!Blood Flows (fraction of cardiac output from Brown)

 CONSTANT QCC = 14.6 ! Cardiac output (L/hr/kg)
2

 CONSTANT QPC = 17.0 ! Alveolar ventilation (L/hr/kg)

 CONSTANT QBoneC = 0.122 ! Bone

 CONSTANT QBrnC = 0.020 ! Brain

 CONSTANT QLivC = 0.183 ! Liver 

 CONSTANT QNoseC = 0.01 ! Nose (Frederick TAP 1998)

!Biliary elimination flow and rates 

 CONSTANT Kbilec = 0.0054 ! First order rate constant (/hr/kg)

 CONSTANT QBile = 0.937 ! Bile flow (mL/hr)

 CONSTANT kbel = 0.937 ! bile excretion rate

!Fractional issue Volumes (fraction of body weight from Brown)

 CONSTANT BW = 0.325 ! Body weight (kg)

 CONSTANT VBldC = 0.0676 ! Blood

 CONSTANT VBoneC = 0.073 ! Bone

 CONSTANT VBrnC = 0.006 ! Brain

 CONSTANT VLivC = 0.034 ! Liver

 CONSTANT VLungC = 0.005 ! Lung

! Fractional brain region volume (Dorman ToxSci 2001)

 CONSTANT VSTMC = 0.04 ! striatum

 CONSTANT VCBMC = 0.12 ! cerebellum

 CONSTANT VOFBC = 0.03 ! olfactory bulb

 CONSTANT VPITC = 0.0004 ! pituitary (monkey)

 CONSTANT VBILE = 10. ! gallbladder vol (monkey)

! Nasal surface areas (Schroeter ToxSci 2008)

 CONSTANT SANolf = 25.2 ! surface area nasal olfactory (cm2/kg)

 CONSTANT SANresp = 18.0 ! surface area nasal respiratory (cm2/kg)

 CONSTANT THnose = 84. ! thickness nasal tissue (um)
3

!************************************

! Chemical specific parameters

!************************************

! Fractional depositions and clearances (estimated from MPPD and Brenneman ToxSci 2006)

CONSTANT fdepLU = 0.15 ! regional deposition in lung (pulmonary + tracheabronchiol)

CONSTANT fdepNO = 0.08 ! regional deposition in nasal cavity (olfactory)

CONSTANT fdepNR = 0.47 ! regional deposition in nasal cavity (respiratory)

CONSTANT kdepluC = 2.0 ! transport from epithelium to lung tissue (/hr)

CONSTANT kshalluC = 100.0 ! systemic uptake from lung epithelium (/hr)

CONSTANT kdepNRc = 0.041 ! systemic uptake from nasal respiratory (/hr)

CONSTANT kNPOBC = 0.012 ! transport from olfactory to brain (/hr)

!Partition Coefficients (assume partition is regulated by binding)

CONSTANT PLIV = 0.85 ! blood:liver PC

CONSTANT PBODY = 1.0 ! blood: others PC

CONSTANT PBONE = 0.5 ! blood: bone PC

CONSTANT PLUNG = 1.0 ! blood: lung PC

CONSTANT PBILE = 1.0 ! blood: bile PC

! Brain permeablity diffusion flux constants (fitted with data)

CONSTANT kinstc = 1.7 ! striatum influx rate constant (/hr/kg)

CONSTANT kincbc = 1.0 ! cerebellum influx rate constant (/hr/kg)

CONSTANT kinobc = 0.25 ! olfactory bulb influx rate constant (/hr/kg)

CONSTANT kinptc = 0. ! pituitary influx rate constant (/hr/kg)

CONSTANT koutstc = 1.0 ! striatum efflux rate constant (/hr/kg)

CONSTANT koutcbc = 8.0 ! cerebellum efflux rate constant (/hr/kg)

CONSTANT koutobc = 2.5 ! olfactory efflux rate constant (/hr/kg)
4

CONSTANT koutptc = 0.0 ! pituitary efflux rate constant (/hr/kg)

!Tissue binding constants (fitted with data)

CONSTANT kdliv = 0.011 !Dissociation constant for Liver (/hr)

CONSTANT kdoth = 0.00464 !Dissociation constant for Others (/hr)

CONSTANT kdbone = 0.0238 !Dissociation constant for bone (/hr)

CONSTANT kdlung = 0.38 !Dissociation constant for lung (/hr)

CONSTANT kaliv = 0.277 !Association constant for Liver (/ug/hr)

CONSTANT kaoth = 0.00927 !Association constant for Others (/ug/hr)

CONSTANT kabone = 0.00464 !Association constant for bone (/ug/hr)

CONSTANT kalung = 0.1 !Association constant for lung (/ug/hr)

CONSTANT kdst = 0.001 !Dissociation constant for striatum (/hr)

CONSTANT kdob = 0.0003 !Dissociation constant for olfactory bulb (/hr)

CONSTANT kdcb = 0.00013 !Dissociation constant for cerebellum (/hr)

CONSTANT kdpt = 0. !Dissociation constant for pituitary (/hr)

CONSTANT kast = 0.0258 !Association constant for striatum (/ug/hr)

CONSTANT kaob = 0.60 !Association constant for olfactory bulb (/ug/hr)

CONSTANT kacb = 1.324 !Association constant for cerebellum (/ug/hr)

CONSTANT kapt = 0. !Association constant for pituitary (/ug/hr)

! Maximal storage tissue capacities (fitted with data, ug/g tissue)

CONSTANT BmaxLivc = 2400. !liver

CONSTANT BmaxBodc = 200. !others

CONSTANT BmaxBonec = 600. !bone

CONSTANT BmaxLungc = 14600. !lung

CONSTANT BmaxSTc = 2000. !striatum

CONSTANT BmaxOBc = 1900. !olfactory bulb

CONSTANT BmaxCBc = 1000. !cerebellum

CONSTANT BmaxPTc = 1.e-6 !pituitary
5

! Initial ratio of bound vs free Mn in tissues -- assumes all is mostly bound

CONSTANT fBr = 0.95 !brain

CONSTANT FLiv = 0.9 !liver

CONSTANT FOth = 0.9 !Others

CONSTANT fbone = 0.9 !bone

CONSTANT Flung = 0.9 !lung

CONSTANT fstm = 0.99 !striatum

CONSTANT fofb = 0.9 !olfactory bulb

CONSTANT fcbm = 0.9 !cerebellum

CONSTANT fpit = 0.9 !pituitary

!Basal tissue Mn Concentrations (ug/L) Teeguarden III JTEH 2008

CONSTANT CLiv0 = 2300. !liver 

CONSTANT CBRn0 = 450. !brain 

CONSTANT CSTM0 = 600. !striatum 

CONSTANT COFB0 = 1050. !olfactory bulb

CONSTANT CCBM0 = 500. !cerebellum 

CONSTANT CPIT0 = 0. !pituitary

CONSTANT COth0 = 110. !others 

CONSTANT CART0 = 10. !arterial blood

CONSTANT CVEN0 = 10. !venous blood 

CONSTANT CBody0 = 110.0 !Body (Malecki 1995)

CONSTANT CBone0 = 500.0 !Bone (Davis 1992)

CONSTANT CLung0 = 400.0 !Lung (Wieczorek 1989)

CONSTANT CBile0 = 1.0 !Bile (ug/g)

! Dose-dependant induction constants (fitted with data)

CONSTANT KBinduc = 1.e-6 ! maximal biliary induction factor (0 off or x times added)
6

CONSTANT Km = 3000. ! biliary affinity rate constant

CONSTANT n = 0. ! slope factor induction

CONSTANT kinmax = 0. ! maximal striatal influx induction factor

CONSTANT kipmax = 0. ! maximal pituitary influx induction factor

CONSTANT kmi = 1e-10 ! striatal affinity rate constant

CONSTANT kmp = 1e-10 ! pituitary affinity rate constant

!************************************

! Parameter scaling calculations

!************************************

! Tissue volumes (L)

Vbld=VBLDC*BW ! whole blood (arterial/venous)

VBone = VBoneC * BW ! bone

VLiv = VLivC * BW ! liver

VLung = VLungC * BW ! lung

VNolf = SANolf * (BW**0.75) * THnose / 10000 / 1000 ! nasal olfactory region 

VNresp = SANresp * (BW**0.75) * THnose / 10000 / 1000 ! nasal respiratory region

VNose = VNolf + VNresp ! nasal cavity

VBrn = VBrnC * BW ! whole brain

VSTM = VSTMC * VBRN ! striatum

VOFB = VOFBC * VBRN ! olfactory bulb

VCBM = VCBMC * VBRN ! cerebellum

VPIT = VPITC * VBRN ! pituitary

IF (VPITC .LT. 1.e-6) THEN ! switch for the pituitary

 spit = 0. 

ELSE 
7

 spit = 1. 

ENDIF ! turn pituitary off (0) for rat or on (1) for monkey

VOTH = BW-VSTM-VOFB-VCBM-VBRN*0.03-VLIV &

-VBONE-VLUNG-VBLD-VNOSE-(VPIT*spit) ! Other tissues

! Tissue blood flows (L/hr)

QC = QCC*BW**0.75 ! Cardiac Output

QP = QPC*BW**0.75 ! Alveolar ventilation

QL = QLIVC*QC ! liver

QBR = QBRNC*QC ! brain

QBONE = QBONEC*QC ! bone

Qnose = Qnosec*QC ! nose

QO = QC-QBONE-QBR-QL-QNose ! Other tissues

! Scale biliary excretion to BW^.75

Kbile = Kbilec * (BW**(0.75)) !Biliary elimination (L/hr)

! Scale nasal and lung epithelium clearances to BW^.25

Kdeplu = kdepluc*(BW**(-0.25)) !Deposited lung clearance to deep tissues (/hr)

Kshallu = kshalluc*(BW**(-0.25)) !Deposited lung clearance to deep tissues (/hr)

kdepNR = kdepNRc*(BW**(-0.25)) !Deposited nose clearance to blood (/hr)

kNPOB = kNPOBc*(BW**(-0.25)) !nose clearance to brain (/hr) 

! Scale influx/efflux rates to BW^.25

kinst = kinstc * (BW**(-0.25)) ! striatum influx (/hr)

kinob = kinobc * (BW**(-0.25)) ! olfactory bulb influx (/hr)

kincb = kincbc * (BW**(-0.25)) ! cerebellum influx (/hr)

kinpt = kinptc * (BW**(-0.25)) ! pituitary influx (/hr)
8

koutst = koutstc * (BW**(-0.25)) ! striatum efflux (/hr)

koutob = koutobc * (BW**(-0.25)) ! olfactory bulb efflux (/hr)

koutcb = koutcbc * (BW**(-0.25)) ! cerebellum efflux (/hr)

koutpt = koutptc * (BW**(-0.25)) ! pituitary efflux (/hr)

!**********************************************

! Background tissue Mn level calculations

!**********************************************

! Initial tissue levels

AL0 = CLiv0*VLIV !Total background amount in liver (ug)

ABR0 = CBRn0*(VBRN*0.03) !Total background amount in brain (ug)

AO0 = COth0*VOTH !Total background amount in others (ug)

ABone0 = Cbone0*Vbone !Total background amount in bone (ug)

ALUNG0 = Clung0*Vlung !Total background amount in lung (ug)

AVEN0 = CVEN0*VBLD !Total background amount in venous (ug)

AART0 = CART0*VBLD !Total background amount in arterial (ug)

ASTM0 = CSTM0*VSTM !Total background amount in striatum (ug)

AOFB0 = COFB0*VOFB !Total background amount in olfactory bulb (ug)

ACBM0 = CCBM0*VCBM !Total background amount in cerebellum (ug)

APIT0 = CPIT0*VPIT !Total background amount in cerebellum (ug)

! Tissue maximal capacities

BmaxLiv = BmaxLivc * VLIV !liver

BmaxBod = BmaxBodc * VOTH !others

BmaxBone = BmaxBonec * Vbone !bone

Bmaxlung = Bmaxlungc * Vlung !lung

BmaxST = BmaxSTc * VSTM !striatum
9

BmaxOB = BmaxOBc * VOFB !olfactory bulb

BmaxCB = BmaxCBc * VCBM !cerebellum

BmaxPT = BmaxPTc * VPIT !pituitary

! Basal stored (bound) Mn capacities (ug) -- assumes all is mostly bound

BALiv0 = AL0*fLiv !liver

BABody0 = AO0*foth !others

BABone0 = ABone0*fbone !bone

BALung0 = Alung0*flung !lung

BABr0s = ASTM0*fstm !striatum

BABr0o = AOFB0*fofb !olfactory bulb

BABr0c = ACBM0*fcbm !cerebellum

BABr0p = APIT0*fpit !pituitary

! Basal free Mn in tissues (ug)

FABr0b = ABR0*(1-fBr) !brain

FALiv0 = AL0*(1-fLiv) !liver

FABody0 = AO0*(1-foth) !others

FABone0 = ABone0*(1-fbone) !bone

FALung0 = Alung0*(1-flung) !lung

FABr0s = ASTM0*(1-fstm) !striatum

FABr0o = AOFB0*(1-fofb) !olfactory bulb

FABr0c = ACBM0*(1-fcbm) !cerebellum

FABr0P = APIT0*(1-fpit) !pituitary

! Free tissue capacities

BLiv = Bmaxliv - BALiv0 !liver

BOth = BmaxBod - BABody0 !others

Bbone = BmaxBone - BABone0 !bone

Blung = Bmaxlung - BALung0 !lung
10

BST = BmaxST - BABr0s !striatum

BOB = BmaxOB - BABr0o !olfactory bulb

BCB = BmaxCB - BABr0c !cerebellum

BPT = BmaxPT - BABr0p !pituitary

!********************************

! Dosing Controls 

!********************************

! Simulation controls

CONSTANT TSTOP = 120. ! end simulation (days)

Tsim = TSTOP*24. ! end simulation (hours)

CONSTANT MAXT = 1. ! maximum tolerance

CONSTANT CINTC = 1.0 ! comm. interval

CINT = CINTC ! default comm interval

! Dietary inputs

CONSTANT DDIET = 0. ! Diet (ppm)

CONSTANT BDIET = 0. ! Diet (mg/day)

CONSTANT Fdietup = 0.008 ! Diet absorption fraction

CONSTANT InFac = 0.05 ! Dietary Intake Factors (kg/day/BW^0.75) (from EPA 1986?)

CONSTANT xdoff = 12000000. ! switch to turn diet off at day X

xdin = 1. ! Diet on by default

CONSTANT DORAL = 0. ! oral bolus like tracer (mg)

constant KENT =0.03

constant kGi=0.99

constant Fent = 0.01
11

! Inhalation inputs

CONSTANT DINH = 0. ! Inhaled dose (mg/m3)

CONSTANT tlen = 0.1 ! length of inhaled exposure (hours)

CONSTANT Dstart = 0. ! First day of inhalation (day)

CONSTANT Dstop = 0. ! Last day of inhalation (day)

CONSTANT DaysOn = 0. ! How many days exposed per week

DOn = DaysOn*24. ! days on (hours)

Doff = (Dstop - Dstart)*24. ! stop (hours)

TimeOn = (Dstart*24.0) ! start (hours)

TimeOff = (Dstop*24.0) + tlen ! next exposure (hours)

Weekly = 0. ! week schedule

Daily = 0. ! day schedule

Exposure = 0. ! lenght of dosing period (days)

! IV tracer (54Mn)

CONSTANT DIVR = 0. ! IV tracer injection dose (uCi)

CONSTANT SA = 401. ! Specific activity (uCi/ug)

CONSTANT IVRon = 0.0 ! first day of IV tracer (day)

CONSTANT LendoseR = 0.01 ! Length of IV (hrs)

CONSTANT FBq = 37. ! Conversion Factor (kBq/uCi)

CONSTANT khalf = 312. ! Mn54 half life (days)

kloss = 0.693/(khalf*24) ! Mn54 decay (hrs-1)

IVR = IVRon *24. ! start (hours) 

TimeOffR = IVRon*24.0+LenDoseR ! next exposure (hours)

CONSTANT DIV = 0. ! cold Mn (ug)

! oral tracer (54Mn) 

CONSTANT DDIETR = 0. ! dose (uCi)

CONSTANT Kfeces = 0. ! fecal excretion rate (/hr)

CONSTANT Kurine = 0. ! urinary excretion rate (/hr)
12

xdiet = DDIETR/SA

!CONSTANT KGI = 0.2 ! GI tract absorption delay

CONSTANT KIL = 0.

! oral tracer (54Mn) 

! inhaled tracer (54Mn) 

CONSTANT DINHR = 0.

CONSTANT fdepx = 1.

xCINH = DINHR / SA

! inhaled tracer (54Mn) 

! subcutaneous tracer (54Mn) 

 CONSTANT DSUBQ = 0.

 CONSTANT KSBQ = 0.

 CONSTANT ksqr = 0.012

 xsubq = DSUBQ / SA

 CONSTANT SUBQ = 0. ! mixed cold Mn (ug)

! subcutaneous tracer (54Mn) 

CONSTANT DIPR = 0.0 ! Radio-labeled IP injection dose (uCi)

CONSTANT KIPSLOW = 0.0 ! parameter for slow release from peritoneum to liver

CONSTANT IPFRAC = 0.05; ! fraction of IP dose to slow IP compartment

xipr = DIPR/SA 

 ! use IVR, LendoseR, and TimeoffR from iv injection

! Initialize input rates

kDiet = 0. ! daily dietary intake (ug/hr)

kInh = 0. ! inhaled input rate (ug/hr)
13

kIVR = 0. ! iv 54Mn input rate (ug/hr)

kdietr = 0.

kinhr = 0.

ksubqr = 0.

ksubq = 0.

kiv = 0.

koral = 0.

kipr = 0.

END !Initial

DYNAMIC 

ALGORITHM IALG = 2 !Gear algorithm

!********************************

! Dosing calculations

!********************************

! Mn Diet uptake

kDiet = (BDIET+(DDIET*InFac*BW**0.75))*1000/24 !Dose in ug /hr

xdin = PULSE(0.0, Tsim , xdoff) !switch diet on or off 

! Mn Inhalation

CINH = DINH !Dose in ug/L

DISCRETE DoseOn ! exposure on

 INTERVAL DoseInt = 24.0

 SCHEDULE DoseOff .AT. T + tlen
14

 IF ((T.GE.TimeOn) .AND. (T.LE.TimeOff)) THEN

 IF (DINH.GT.0.0) CINT = 0.1

 kInh = QP * CINH * fdepLU

 kInhNO = QP * CINH * fdepNO

 kInhNR = QP * CINH * fdepNR

ksubq = subq /LendoseR !* ksqr

 IF (TLEN.GT.0) kiv = DIV/tlen

 IF (TLEN.GT.0) koral = DORAL*1000/tlen

 

 ENDIF

END

DISCRETE DoseOff !stop exposure

 CINT = CIntC

 kInh = 0.0

 kInhNO = 0.0

 kInhNR = 0.0

 ksubq = 0.

 kiv = 0.0

 TimeOff = TimeOff + 24.0

END

Daily = PULSE(TimeOn, 24., tlen) !Daily scheduling

Weekly = PULSE(TimeOn , 168. , DOn) !Weekly scheduling

Exposure = Daily * Weekly * PULSE(TimeOn, Tsim , Doff) !exposure switch for week period

! 54Mn tracer uptake

DISCRETE DoseROn ! tracer exposure on

 INTERVAL DoseRInt = 24.0
15

 SCHEDULE DoseROff .AT. T + LenDoseR

 IF ((T.GE.IVR) .AND. (T.LE.TimeOffR)) THEN

 IF (DIVR.GT.0.0) CINT = LenDoseR/10

 IF (DIVR.GT.0.0) kivr = (DIVR/SA)/LenDoseR

 IF (DIPR.GT.0.0) kipr = (DIPR/SA)/LenDoseR

 IF (DdietR.GT.0.0) kdietr = xdiet/LenDoseR

 IF (DINHr.GT.0.0) kinhr = QP * xCINH * fdepx

 IF (DsubQ.GT.0.0) ksubqr = xsubq /LendoseR !* ksqr

 ENDIF

END

DISCRETE DoseROff !stop tracer exposure

 CINT = CIntC

 kivr = 0.

 kdietr = 0.

 kinhr = 0.

 ksubqr = 0.

 kipr = 0.

END

! convert hours into days

Days = T/24.0 - Dstart

DERIVATIVE 

!***************************************

! Model differential Equations

!***************************************
16

! ******* LUNG ******* 

!Adeplu = amount deposited into lungs (ug)

RAdeplu = kinh*Exposure - kdeplu * Adeplu - kshallu * Adeplu ! Mn on lung epithelium over time (ug/hr)

Adeplu = INTEG(RAdeplu, 0.0) ! Mn on lung epithelium (ug)

!Alung = Amount in blood LUNG(ug)

RAlung = QC*(CV - CVlung) + kdlung*BMnlung - &

 kalung*Blung*AVlung + kdeplu * Adeplu ! Free Mn in lung over time (ug/hr)

Alung = INTEG( RAlung, FAlung0) ! Mn in lung blood (ug)

AVlung = Alung / Plung ! Mn leaving the lung into venous system (ug)

CVlung = AVlung / Vlung ! Free lung Mn concentration (ug/kg)

!BMnlung = Amount bound in LUNG(ug)

RBMnlung = - kdlung * BMnlung + kalung * Blung * AVlung ! Mn stored in lung over time (ug/hr)

BMnlung = INTEG(RBMnlung ,BAlung0) ! Mn stored in lung (ug)

CBMnlung = BMnlung / Vlung ! Bound lung Mn concentration (ug/kg)

!Blung = capacity in LUNG(ug)

Blung = Bmaxlung - BmnLung - xBmnLung ! Free tissue capacity (ug)

PCBlung = Blung / Bmaxlung * 100. ! % Free capacity 

! Atotlung = Total amount of Mn in LUNG(ug)

Atotlung = Alung + BMnlung + ADeplu ! ug

Ctotlung = Atotlung/(Vlung*1000.) ! ug /g tissue

! ******* NOSE ******* 

! Deposited nasal respiratory
17

RADepNR = kinhNR * Exposure - kdepNR * AdepNR ! Mn on nasal cavity over time (ug/hr)

ADepNR = integ(RAdepNR, 0.0) ! Mn on nasal cavity (ug) 

! Nasal respiratory blood

RANose = Qnose*(CA - CVNose) + kdepNR * AdepNR ! Mn on respiratory epithelium over time (ug/hr)

ANose = integ(RANose, 0.0) ! Mn on respiratory epithelium (ug)

CVNose = ANose / VNRESP ! respiratory Mn concentration (ug/kg)

! Deposited olfactory region

RANolf = kinhNO * Exposure - kNPOB * ANolf ! Mn on ofactory epithelium over time (ug/hr)

ANolf = integ(RANolf, 0.0) ! Mn on ofactory epithelium (ug)

CVNolf = ANolf / VNOLF ! olfactory Mn concentration (ug/kg)

AtotNose = ANose + ADepNR + ANolf ! ug

CtotNose = Atotnose/(Vnose*1000.) ! ug /g tissue

! ******* LIVER *******

kbilex = kbile *(1. + kbinduc*(CART+xCART)**n / &

 (km**n+(CART+xCART)**n)) ! free Mn dependant excretion rate constant (/hr)

RBile = kbilex*CVL ! biliary excretion rate (ug/hr)

rloss = Qbile*Cbile ! bile flow elimination

RAbile = rbile - rloss ! amount of Mn in bile over time (ug/hr)

Abile = INTEG(RAbile, 0.0) ! amount of Mn in bile (ug)

Cbile = Abile/Vbile ! bile Mn concentration (ug/g)
18

! Free Mn in liver 

RAL = QL*(CA - CVL) + kdliv*BMnLiv - kaliv*BLiv*AVL & 

 - rbile +RADI ! Free Mn in liver over time (ug/hr)

AL = INTEG(RAL, FALiv0) ! Mn in liver blood (ug)

AVL = AL/PLIV ! Mn leaving the liver into venous system (ug)

CVL = AVL/VLIV ! Free Mn concentration (ug/kg)

!Bound Mn in liver

RBMnLiv = - kdliv * BMnLiv + kaliv * BLiv * AVL ! Mn stored in tissue over time (ug/hr)

BMnLiv = INTEG(RBMnLiv , BALiv0) ! Mn stored in tissue (ug)

CBMnLiv = BMnLiv / VLIV ! Bound Mn concentration (ug/kg)

!Liver tissue capacity

BLiv = Bmaxliv - BmnLiv - xBmnLiv ! Free tissue capacity (ug)

PCBLiv = BLiv / Bmaxliv * 100. ! % Free capacity 

! Total Mn level in Liver

AtotLiv = AL + BMnLiv ! ug

CtotLiv = AtotLiv/(VLIV*1000.) ! ug /g tissue

!GUT 

RADI = kdiet*xdin*fdietup ! rate of dietary Mn input to liver

ADI = INTEG (RADI,0.0)

RAGL = kdiet*xdin*(1-fdietup) - (1-Fent)*kGi*AGL - Fent*kGi*AGL ! rate of change in gut lumen

AGL = integ (RAGL,0.0) ! Amount of Mn in gut lumen (not available to systemic, not uptaken)

RAGE = Fent*kgi*AGL -kEnt*AGE ! rate of Mn in GI epithelium

 ! kEnt = slow loss from enterocytes to feces

AGE = integ (RAGE,0.0) ! Amount of Mn in enterocytes 
19

RALGL = (1-Fent)*kGi*AGL - Rfeces + Rbile +kEnt*AGE

ALGL = integ (RALGL,0.0) ! Amount of Mn in lower GI

Rfeces = kfeces*ALGL ! ug Mn/hr

rfecesmg = rfeces/1000/0.5 ! mg Mn/g feces, assuming 12g/day fecal volume)

Feces = integ (Rfeces,0.0) ! Amount in feces

! ******* BRAIN ******* 

! brain dose-dependent pump

kinstx = kinst *(1. + kinmax*abrst / (kmi+abrst)) ! free Mn dependant striatal inlfux constant (/hr)

kinptx = kinpt *(1. + kipmax*abrpt / (kmp+abrpt)) ! free Mn dependant pituitary inlfux constant (/hr)

!Free Mn in brain blood supply

RABrBd = QBR*(CA - CVBr) + koutst*ABrst + &

 koutob*ABrob + koutcb * ABrcb + koutpt * ABrpt &

 - (kinstx + kinob + kincb + kinptx) * ABrBd ! Free Mn in brain blood over time (ug/hr)

ABrBd = INTEG(RABrBd , FABr0b) ! Free Mn in brain blood (ug)

CBrb = ABrBd / (VBRN*0.03) ! based on blood volume in the brain (ug/kg)

CVBr = CBrB ! Free Mn concentration leaving the brain (ug/kg)

!Free Mn in brain tissue (ug) - STRIATUM

RABrST = kinstx*ABrBd - koutst*ABRST + kdst*BMnST &

 - kast*BST*ABrST ! Free Mn in STM over time (ug/hr)

ABrST = INTEG(RABrST , FABr0s) ! Free Mn in STM (ug)

CBrST = ABrST / VSTM ! Free STM Mn concentration (ug/kg)

!Bound Mn in brain tissue (ug) - STRIATUM

RBMnST = - kdst*BMnST + kast*BST*ABrST ! Mn stored in STM over time (ug/hr)

BMnST = INTEG(RBMnST , BAbr0s) ! Mn stored in STM (ug)
20

CBMnST = BMnST / VSTM ! Bound STM Mn concentration (ug/kg)

!Mn capacity in STM (ug)

BST = BmaxST - BmnST - xBmnST ! Free STM capacity (ug)

PCBST = BST / BmaxST * 100. ! % Free capacity 

! total amount in STM

AST = ABrST + BMnST + ABrbd*VSTMC ! ug

CtotST = AST / VSTM /1000. ! ug /g tissue

!Free Mn in brain tissue (ug) - OLFACTORY BULB

RABrOB = kinOB*ABrBd - koutOB*ABROB + kdob*BMnOB &

 - kaob*BOB*ABrOB + kNPOB * ANolf ! Free Mn in OFB over time (ug/hr)

ABrOB = INTEG(RABrOB , FABr0o) ! Free Mn in OFB (ug)

CBrOB = ABrOB / VOFB ! Free OFB Mn concentration (ug/kg)

!Bound Mn in brain tissue (ug) - OLFACTORY BULB

RBMnOB = - kdob*BMnOB + kaob*BOB*ABrOB ! Mn stored in OFB over time (ug/hr)

BMnOB = INTEG(RBMnOB , BAbr0o) ! Mn stored in OFB (ug)

CBMnOB = BMnOB / VOFB ! Bound OFB Mn concentration (ug/kg)

!Mn capacity in Olfactory bulb (ug)

BOB = BmaxOB - BmnOB - xBmnOB ! Free STM capacity (ug)

CBOB = BOB / BmaxOB * 100. ! % Free capacity 

! total - OFB

AOB = (ABrOB + BMnOB) + ABrbd*VOFBC ! ug

CtotOB = (AOB) / VOFB /1000. ! ug /g tissue
21

!Free Mn in brain tissue (ug) - CEREBELLUM

RABrCB = kinCB*ABrBd - koutCB*ABRCB + kdcb*BMnCB - &

 kacb*BCB*ABrCB ! Free Mn in CBM over time (ug/hr)

ABrCB = INTEG(RABrCB , FABr0c) ! Free Mn in CBM (ug)

CBrCB = ABrCB / VCBM ! Free CBM Mn concentration (ug/kg)

!Bound Mn in brain tissue (ug) - CEREBELLUM

RBMnCB = - kdcb*BMnCB + kacb*BCB*ABrCB ! Mn stored in CBM over time (ug/hr)

BMnCB = INTEG(RBMnCB , BAbr0c) ! Mn stored in CBM (ug)

CBMnCB = BMnCB / VCBM ! Bound CBM Mn concentration (ug/kg)

!Mn capacity in CBM (ug)

BCB = BmaxCB - BmnCB - xBmnCB ! Free CBM capacity (ug)

PCBCB = BCB / BmaxCB * 100. ! % Free capacity 

! total in CBM

ACB = (ABrCB + BMnCB) + ABrbd*VCBMC ! ug

CtotCB = (ACB) / VCBM /1000. ! ug /g tissue

!Free Mn in brain tissue (ug) - PITUITARY

RABrPT = kinPTx*ABrBd - koutPT*ABRPT + kdpt*BMnPT &

 - kapt*BPT*ABrPT ! Free Mn in PIT over time (ug/hr)

ABrPT = INTEG(RABrPT , FABr0p) ! Free Mn in PIT (ug)

CBrPT = ABrPT / VPIT ! Free PIT Mn concentration (ug/kg)

!Bound Mn in brain tissue (ug) - PITUITARY

RBMnPT = - kdpt*BMnPT + kapt*BPT*ABrPT ! Mn stored in PIT over time (ug/hr)

BMnPT = INTEG(RBMnPT , BAbr0p) ! Mn stored in PIT (ug)
22

CBMnPT = BMnPT / VPIT ! Bound PIT Mn concentration (ug/kg)

!Mn capacity in PIT (ug)

BPT = BmaxPT - BmnPT - xBmnPT ! Free PIT capacity (ug)

PCBPT = BPT / BmaxPT * 100. ! % Free capacity 

! total in PIT

APT = (ABrPT + BMnPT) + ABrbd * VPITC * spit ! ug

CtotPT = (APT) / VPIT /1000. ! ug /g tissue

! AtotBr = Total amount of Mn in Brain tissues(ug)

ABr = ABrST + ABrOB + ABrCB + ABrPT ! Free Mn in brain regions (ug/hr)

BMnBr = BMnST + BMnOB + BMnCB + BMnPT ! Mn stored in brain regions (ug)

CProtMnBr = BMnBr / (VSTM+VOFB+VCBM+VPIT) ! Bound brain region Mn concentration (ug/kg)

AtotBr = ABr + BMnBr + ABrBd ! Total brain Mn incl. blood (ug)

CtotBr = AtotBr/((VSTM+VOFB+VCBM+VPIT+VBRN*0.03)*1000.) ! ug /g tissue

! ******* BONE *******

!Abone = Amount free in BONE(ug)

RAbone = QBONE*(CA - CVbone) + kdbone*BMnbone &

 - kabone*Bbone*AVbone ! Free Mn in others over time (ug/hr)

Abone = INTEG( RAbone, FAbone0) ! Mn in others blood (ug)

AVbone = Abone / Pbone ! Mn leaving out to venous system (ug)

CVbone = AVbone / Vbone ! Free Mn concentration (ug/kg)

!BMnbone = Amount of bound in BONE(ug)

RBMnbone = - kdbone * BMnbone + kabone * Bbone * AVbone ! Mn stored in tissue over time (ug/hr)
23

BMnbone = INTEG(RBMnbone ,BAbone0) ! Mn stored in tissue (ug)

CBMnbone = BMnbone / Vbone ! Bound Mn concentration (ug/kg)

!Bbone = Capacity BONE(ug)

Bbone = BmaxBone - BmnBone - xBmnBone ! Free tissue capacity (ug)

PCBbone = Bbone / BmaxBone * 100. ! % Free capacity 

! Atotbone = Total amount of Mn in BONE(ug)

Atotbone = Abone + BMnbone ! ug

Ctotbone = Atotbone/(Vbone*1000.) ! ug /g tissue

! ******* REST OF BODY (OTHERS) *******

!Free Mn in others/body

RAO = QO*(CA - CVO) + kdoth*BMnOth - kaoth*BOth*AVO ! Free Mn in others over time (ug/hr)

AO = INTEG( RAO, FABody0) ! Mn in others blood (ug)

AVO = AO / PBODY ! Mn leaving out to venous system (ug)

CVO = AVO / VOTH ! Free Mn concentration (ug/kg)

!Bound Mn in others 

RBMnOth = - kdoth * BMnOth + kaoth * BOth * AVO ! Mn stored in tissue over time (ug/hr)

BMnOth = INTEG(RBMnOth ,BABody0) ! Mn stored in tissue (ug)

CBMnOth = BMnOth / VOTH ! Bound Mn concentration (ug/kg)

!Other tissue capacity

BOth = BmaxBod - BmnOth - xBmnOth ! Free tissue capacity (ug)

PCBOth = BOth / BmaxBod * 100. ! % Free capacity 
24

! AtotOth = Total amount of Mn in Others (ug)

AtotOth = AO + BMnOth ! ug

CtotOth = AtotOth/(VOTH*1000.) ! ug /g tissue

!BLOOD

!Free Mn arterial blood concentration (ug) 

RAArt = QC * CVLUNG - QC * CA + kshallu * Adeplu 

AArt = INTEG(RAArt , AArt0)

CA = AArt/VBLD 

CART = AArt/VBLD/1000. ! ug /g tissue

!FRee Mn venous blood concentration (ug) 

RAVen = QL*CVL + QBR*CVBr + QO*CVO + QBONE* CVbone - &

 QC*CV + Qnose * CVNose + kiv*exposure +ksubq*exposure!+ ksbq * ASQ

AVen = INTEG( RAVen , AVen0 )

CV = AVEN/VBLD

CVEN = AVen/VBLD/1000. ! ug /g tissue

!Subcutaneaous 

RASQ = ksubq * exposure - ksbq * ASQ 

ASQ = INTEG(RASQ, 0.0)

doseQabsorbed=integ(ksbq*asq,0)

doseQinput = integ(ksubq*exposure,0)

!***********************************************************************

! Tracer Kinetics
25

!***********************************************************************

!LUNG

! lung depot and inhalation 

! xAdeplu = amount deposited into lungs (ug)

xRAdeplu = kinhr - kdeplu*xAdeplu - kshallu*xAdeplu ! Mn on lung epithelium over time (ug/hr)

xAdeplu = INTEG(xRAdeplu, 0.0) ! Mn on lung epithelium (ug)

!Alung = Amount in blood LUNG(ug)

xRAlung = QC*(xCV - xCVlung)+ kdlung*xBMnlung - &

kalung*Blung*xAVlung + kdeplu*xAdeplu 

xAlung = INTEG(xRAlung, 0.0)

xAVlung = xAlung / Plung

xCVlung = xAVlung / Vlung

!BMnlung = Amount bound in LUNG(ug)

xRBMnlung = - kdlung*xBMnlung + kalung*Blung*xAVlung 

xBMnlung = INTEG(xRBMnlung , 0.0)

!Atotlung = Total amount of Mn in LUNG(ug)

xAtotlung = xAlung + xBMnlung

! Nasal respiratory blood

xRANose = Qnose*(xCA - xCVNose) 

xANose = integ(xRANose, 0.0)

xCVNose = xANose / VNRESP 

!GUT 
26

xRADI = kdietr*xdin*fdietup ! rate of dietary Mn input to liver

xRAGL = kdietr*xdin*(1.-fdietup) - (1-Fent)*kgi*xAGL - Fent*kGi*xAGL ! rate of change in gut lumen

xAGL = integ (xRAGL,0.0) ! Amount of Mn in gut lumen (not available to systemic, not uptaken)

xRAGE = Fent*kgi*xAGL & ! rate of Mn in GI epithelium

 -kEnt*xAGE ! Fent=fractional retention in enterocytes, kEnt = slow loss from enterocytes to feces

xAGE = integ (xRAGE,0.0) ! Amount of Mn in enterocytes 

xRALGL = (1-Fent)*kGi*xAGL - xRfeces + xRbile +kEnt*xAGE

xALGL = integ (xRALGL,0.0) ! Amount of Mn in lower GI

xRfeces = kfeces*xALGL ! back calculating total Mn amount from hot 

xrfecesmg = (xrfeces/1000)/0.5 

xRfecestotal = SANEWland*xrfeces ! for appraoch 2, ppt file 10/19/09

xrfecesmgtotal = (xrfecestotal/1000)/0.5 ! assuming 12 g/day fecal volume, 12/24=0.5

xFeces = integ (xRfeces,0.0) ! Amount in feces

xfecesmg = integ(xrfecesmg,0)

constant SAnewland = 1000.

!LIVER

xRAL = QL*(xCA - xCVL) + kdliv*xBMnLiv - &

kaliv*BLiv*xAVL - xrbile + XRADI + (1.0-ipfrac)*kipr + kipslow*xAIP

xAL = INTEG(xRAL, 0.0)

xAVL = xAL/PLIV

xCVL = xAVL/VLIV

xrbile = kbilex * xCVL

!BMnLiv = Amount of bound in liver (ug)

xRBMnLiv = - kdliv * xBMnLiv + kaliv*BLiv*xAVL

xBMnLiv = INTEG(xRBMnLiv , 0.0)
27

!AtotLiv = Total amount of Mn in Liver (ug)

xAtotLiv = xAL + xBMnLiv

!BRAIN

!ABrBd = Amount free in brain blood (ug)

xRABrBd = QBR*(xCA - xCVBr) + koutst*xABrst + koutob*xABrob &

+ koutcb * xABrcb + koutpt * xABrpt &

- (kinst + kinob + kincb + kinpt)*xABrBd 

xABrBd = INTEG(xRABrBd , 0.0)

xCBrb = xABrBd / (VBRN*0.03) ! blood volume in brain

xCVBr = xCBrB 

!ABrST = Amount free in brain tissue (ug) - STRIATUM

xRABrST = kinst*xABrBd - koutst*xABRST &

+ kdst*xBMnST - kast*BST*xABrST

xABrST = INTEG(xRABrST , 0.0)

!BMnST = Amount bound in brain tissue (ug) - STRIATUM

xRBMnST = - kdst*xBMnST + kast*BST*xABrST 

xBMnST = INTEG(xRBMnST , 0.0)

! total amount in STM

xAST = xABrST + xBMnST + xABrbd*VSTMC

!ABrOB = Amount free in brain tissue (ug) - OLFACTORY BULB

xRABrOB = kinOB*xABrBd - koutOB*xABROB &

+ kdob*xBMnOB - kaob*BOB*xABrOB

xABrOB = INTEG(xRABrOB , 0.0)

!BMnOB = Amount bound in brain tissue (ug) - OLFACTORY BULB
28

xRBMnOB = - kdob*xBMnOB + kaob*BOB*xABrOB

xBMnOB = INTEG(xRBMnOB , 0.0)

! total - OFB

xAOB = (xABrOB + xBMnOB) + xABrbd*VOFBC 

!ABrCB = Amount free in brain tissue (ug) - CEREBELLUM

xRABrCB = kinCB*xABrBd - koutCB*xABRCB &

+ kdcb*xBMnCB - kacb*BCB*xABrCB

xABrCB = INTEG(xRABrCB , 0.0)

!BMnCB = Amount bound in brain tissue (ug) - CEREBELLUM

xRBMnCB = - kdcb*xBMnCB + kacb*BCB*xABrCB 

xBMnCB = INTEG(xRBMnCB , 0.0)

! total in CBM

xACB = (xABrCB + xBMnCB) + xABrbd*VCBMC

!ABrPT = Amount free in brain tissue (ug) - PITUITARY

xRABrPT = kinPT*xABrBd - koutPT*xABRPT &

+ kdpt*xBMnPT - kapt*BPT*xABrPT 

xABrPT = INTEG(xRABrPT , 0.0)

!BMnPT = Amount bound in brain tissue (ug) - PITUITARY

xRBMnPT = - kdpt*xBMnPT + kapt*BPT*xABrPT 

xBMnPT = INTEG(xRBMnPT , 0.0)

! total in PIT

xAPT = (xABrPT + xBMnPT) + xABrbd*VPITC*spit 

! AtotBr = Total amount of Mn in Brain tissues(ug)

xABr = xABrST + xABrOB + xABrCB + xABrPT 
29

xBMnBr = xBMnST + xBMnOB + xBMnCB + xBMnPT

xAtotBr = xABr + xABrBd + xBMnBr

!BONE

!Abone = Amount free in BONE(ug)

xRAbone = QBONE*(xCA - xCVbone)+ kdbone*xBMnbone - &

kabone*Bbone*xAVbone 

xAbone = INTEG( xRAbone, 0.0)

xAVbone = xAbone / Pbone

xCVbone = xAVbone / Vbone

!BMnbone = Amount of bound in BONE(ug)

xRBMnbone = - kdbone*xBMnbone + kabone*Bbone*xAVbone

xBMnbone = INTEG(xRBMnbone , 0.0)

! Atotbone = Total amount of Mn in BONE(ug)

xAtotbone = xAbone + xBMnbone

!REST OF BODY

!AO = Amount free in others (ug)

xRAO = QO*(xCA - xCVO) + kdoth*xBMnOth &

- kaoth*BOth*xAVO

xAO = INTEG( xRAO, 0.0)

xAVO = xAO / PBODY

xCVO = xAVO / VOTH

!BMnOth = Amount of bound in others (ug)

xRBMnOth = - kdoth*xBMnOth + kaoth*BOth*xAVO

xBMnOth = INTEG(xRBMnOth , 0.0)

! AtotOth = Total amount of Mn in Others (ug)
30

xAtotOth = xAO + xBMnOth

!BLOOD

!CA = Arterial blood concentration (ug) 

xRAArt = QC * xCVLUNG - QC * xCA + kshallu*xAdeplu 

xAArt = INTEG(xRAArt, 0.0)

xCA = xAArt/VBLD

xCART = xAArt/VBLD/1000. ! ug /g tissue

!CV = Venous blood concentration (ug) 

xRAVen = QL*xCVL + QBR*xCVBr + QO*xCVO + QBONE*xCVbone &

- QC*xCV + kivr + Qnose*xCVnose + ksbq * xASQ 

xAVen = INTEG(xRAVen , 0.0 )

xCV = xAVEN/VBLD

xCVEN = xAVen/VBLD/1000. ! ug /g tissue

!Subcutaneaous 

xRASQ = ksubqr* exposure - ksbq * xASQ 

xASQ = INTEG(xRASQ, 0.0)

!IP 

xRAIP = ipfrac*kipr - kipslow*xAIP

xAIP = INTEG(xRAIP, 0.0)

! Tracer Body burden (ug)

Atot54 = xAtotLiv + xAtotBr + xAtotOth + xAArt + &

xAVen + xAtotbone + xAtotlung + xanose &

 + xAdeplu + xASQ + xAGL + xAGE + xALGL + xAIP 
31

! Whole body tracer activity (kBq) -- used to compared with data

Atotmn54 = atot54 * SA* FBQ

!******************************

! Mass balance (ug)

!******************************

! Cold Mn mass

Atotbody = AtotLiv + AtotBr + AtotOth + AArt + AVen &

+ Atotbone + Atotlung + AtotNose + ASQ +AGL + AGE + ALGL!+ AGI + AG ! Body 

burden

Atot0 = AL0 + AO0 + AART0 + AVEN0 + ABONE0 + ALUNG0 + &

FABR0b + ASTM0 + AOFB0 + ACBM0 ! Endogenous Mn

MASS = Atotbody - Atot0 ! Tissue Mass from intake

Loss = feces !INTEG(rbile, 0.0) ! Amount loss in bile

RDose = kDiet + kiv + (ksubq + kinh + kinhNO + kinhNR)*exposure ! Total intake rate 

Dose = INTEG( RDose , 0.0) ! Total Dose

IF (Dose .GT. 0.0) THEN 

Balance = (Atotbody - atot0 + Loss)/(Dose)*100

ELSE 

 Balance = 0.0

ENDIF

! Hot Mn mass

RD54 = kivr + kdietr + ksubqr + kinhr + kipr

Dose54 = INTEG(RD54, 0.0) ! Total Dose

xloss = xfeces ! loss in bile + feces

IF (Dose54 .GT. 0.0) THEN 
32

Balance54 = (Atot54 + xloss)/(Dose54)*100

RET54 = (Atot54 / Dose54)*100

ELSE 

 Balance54 = 0.0

 RET54 = 0.0

ENDIF

TERMT(T.GT.(TStop*24)) !Simulation Finished

END !Derivative

END !Dynamic

END !Program
33

% *************************************************************************

% initials.m file

% intial conditions for adult rat model

% last update 07-24-08

CLIV0=2300.0;CLUNG0= 400.0;COFB0=750.;CSTM0=600.0;CCBM0=500.; % initial tissue conc'

FBR=0.95;FLIV=0.9;FLUNG= 0.9;FOFB=0.9;FSTM=0.99;FCBM=0.9; % fraction bound initialy

% tissue:blood partitions

PBODY=1.;PBONE=1.;PBRN=1.;PLUNG=1.;PLIV=1;PNOSE=1;

% Lung

KSHALLUC= 100.;KDEPLUC=10.; % Depot rates to shallow or deep

KALUNG=0.1; KDLUNG=0.38; % binding rate constants (updated 07-24-08)

BMAXLUNGC=14600; % maximal binding site (updated 07-24-08)

% Nose

FDEPLU= 0.15; % Percent deposition lung (updated 07-24-08)

FDEPNR = 0.47; % Percent deposition nose resp (updated 09-26-08)

FDEPNO = 0.08; % Percent deposition nose olfac (updated 09-26-08)

KNPOBC = 0.012; % rate NO depot to OB

KDEPNRC = 0.041; % rate NR depot to blood

% Liver

BMAXLIVC=2400.; % maximal binding site

PLIV=0.85; % liver:blood partition of free Mn (updated 07-24-08)

KALIV=0.276541; KDLIV=0.01113; % binding rate constants
34

% Bone

BMAXBONEC=600.; % maximal binding site

FBONE=0.9; % initial fraction bound

CBONE0=500; % inital tissue conc'

PBONE=0.5; % partition of free bone blood

KDBONE=0.023840; KABONE=0.00464; % binding rate constants

% "Body" tissues

BMAXBODC = 200.;BMAXBRC = 600.; % maximal binding site

KAOTH=0.009271;KDOTH=0.00464; % binding rate constants

% Cerebellum

KINCBC = 1. ; KOUTCBC=8.; % diffusion rate constants

BMAXCBC=1000.; % maximal binding site

KACB=1.32443;KDCB=0.00013; % binding rate constants

% striatum

KINSTC = 1.7; KOUTSTC=1; % diffusion rate constants

BMAXSTC=2000.; % maximal binding site

CSTM0=600.; % initial conc'

KAST=0.02580;KDST=0.00100; % binding rate constants

%olfactory bulb

% adjusted with olfactory contribution (updated 09-26-08)

COFB0=1050.; % inital tissue conc'

BMAXOBC=1900.; % maximal binding site

KINOBC=0.25;KOUTOBC=2.5; % diffusion rate constants

KAOB=0.6;KDOB=0.0003; % binding rate constants
35

% pituitary

KINPTC=0.; KOUTPTC=0.; % diffusion rate constants

CPIT0=0.; % initial conc'

VPITC=1.e-10; % volume (L)

% Dietary constants

DDIET=125; % Diet (ppm)

KBILEC =0.0054; % Biliary excretion L/hr (updated 07-24-08)

FDIETUP = 0.0027; % Fraction absorbed (updated 07-24-08)

% reinitialize doses

LENDOSER = 0.01;

LENDOSE = 0.6;

DINH =0.;

DINHR = 0.;

DIVR = 0.;

DSUBQ = 0.;

DDIETR = 0.;
36

%*******************************************************************************

% monkey.m file

% set parameters values for monkey physiology

% nasal olfactory and respiratory SA based on fractions in human 

% SA of the nose by Menache 57.5 cm2/ 5 kg or 17.2 cm2/kg^.75

% thickness based on average from Connoly EHP 2000

%Physiological parameters

%Blood Flows (fraction of cardiac output from Davies & Morris 1993)

QCC = 19.5; % Cardiac output (L/hr-1kg)

QPC = 30.0; % Alveolar ventilation (L/hr-1kg)

QBONEC = 0.042; % Bone (Brown Human)

QBRNC = 0.065; % Brain

QLIVC = 0.194; % Liver 

QNOSEC = 0.01; % Nose (Frederick TAP 1998)

QPCr = 17.; % rat alveolar ventilation (Brown et al)

%Tissue Volumes (fraction of body weight from Davies & Morris 1993)

BW = 2.5; % Body weight (kg) (Dorman 2006)

VBLDC = 0.0734; % Blood

VBONEC = 0.12; % Bone (Brown Human)

VBRNC = 0.036; % Brain (Dorman 2006)

VLIVC = 0.027; % Liver

VLUNGC = 0.008; % Lung (Dorman 2006)

VSTMC = 0.0015; % striatum (globus pallidus)

VCBMC = 0.08; % cerebellum

VOFBC = 0.00055; % olfactory bulb

VPITC = 0.0013; % pituitary (Dorman et al 2006)
37

rSANolf = 25.21; % rat surface area nasal olfactory (cm2/kg) Schroeter 2008

rSANresp = 18.08; % rat surface area nasal respiratory (cm2/kg) Schroeter 2008

SANOLF = 1.634; % surface area nasal olfactory (cm2/kg) based on 9.5% of tot SA

SANRESP = 15.566; % surface area nasal respiratory (cm2/kg) Menache et al

THNOSE = 374.; % thickness nasal tissue (um) Conolly 2000

% lung and nasal deposition

%FDEPNO = FDEPNO/rSANolf*SANOLF; % olfactory deposition

%FDEPNR = FDEPNR/rSANresp*SANRESP; % respiratory deposition

%FDEPLU = FDEPLU/QPCr*QPC; % lung deposition

FDEPNO = 0.40*0.1; % 40% nasal deposition -- 10% olfactory

FDEPNR = 0.40*0.9; % 40% nasal deposition -- 90% respiratory

FDEPLU = 0.27; % 27% lung deposition

% initials tissue concentrations (ug/g)

COFB0 = 310.;

CSTM0 = 480.;

CCBM0 = 440.;

CPIT0 = 840.;

CLIV0 = 2490.;

CBONE0 = 130.;

CLUNG0 = 150.;

CART0 = 1.;

CVEN0 = 1.;

COTH0 = 150.;
38

CBILE0 = 0.;

% adjusted rate constants

KNPOBC=0.0035;

KDEPNRC=0.0012;

% brain region rate constants re-adjusted so that each region has 60% bound Mn

% and also tried to keep rate constants near rat levels when possible

% for inter-species consistency

% Jeff Schroeter 4/30/2010

% pituitary

FPIT=0.99;

HP = 0.0;

BMAXPTC=2000;

KAPT=0.3;

KDPT=0.036;

KINPTC=1.29e-1;

KOUTPTC=2.0e-1;

KMP=0.2;

KPTLOSS=0.;

KIPMAX=10.0;

%striatum - globus pallidus

KAST=0.2;

KDST=0.0056;

BMAXSTC=600;

KINSTC=1.6e-1;

KOUTSTC=3.0e-1;

KMI=0.16;
39

KSTLOSS=0;

KINMAX=7.9;

%liver

PLIV=1;

CLIV0=2800;

BMAXLIVC=4000;

KALIV=0.0011;

KDLIV=0.0004;

%lung

PLUNG=1; 

BMAXLUNGC=2700;

KALUNG=0.01;KDLUNG=0.038;

%olfactory bulb

BMAXOBC=700;

KAOB=0.3;

KDOB=0.005;

KINOBC=2.9;

KOUTOBC=11.0;

%cerebellum

BMAXCBC=550;

KACB=0.001;

KDCB=0.0014;

KINCBC=4.3;

KOUTCBC=0.08;

% dietary controls
40

QBILE = 1.8; % scaled from 3.0 bile flow (ml/h) for 5 kg (Davies and Morris 1993)

KBEL = 1.4800;

VBILE = 0.03; % gallblader volume 3 ml

FDIETUP = 1.7e-3;% Diet absorption fraction

KBILEC = 0.051; % First order rate constant (/hr-1kg)

DDIET = 133.;

INFAC = 0.05; % Dietary Intake Factors (kg/day/BW) (from NRC 2003)

% exposure controls

DSTART = 43.; % acclimate for 43 days

DSTOP = 133.; % end exposure

DAYSON = 5.; % 5 days/week

TLEN = 6.; % 6 hours/days

TSTOP = 223.; % end simulation

% induction controls

KBINDUC = 2.5; % induction fold constant

KM = 0.027; % affinity rate constant

N=3; % curve modifier

% GI parameters

FENT = 0.011; 

KGI = 0.6;

KENT = 0.0022;
41

% *****************************************************************************

% human.m file

% set parameters values for human physiology

% nasal olfactory and respiratory SA based on fractions in human 

% SA of the nose by Menache 57.5 cm2/ 5 kg or 17.2 cm2/kg^.75

% thickness based on average from Connoly EHP 2000

%Physiological parameters

%Blood Flows (fraction of cardiac output from Clewell 2001)

QCC = 13.0; % Cardiac output (L/hr-1kg)

QPC = 18.0; % Alveolar ventilation (L/hr-1kg)

QBONEC = 0.042; % Bone (Brown Human)

QBRNC = 0.114; % Brain

QLIVC = 0.227; % Liver 

QNOSEC = 0.01; % Nose (Frederick TAP 1998)

QPCr = 30.; % monkey alveolar ventilation 

%Tissue Volumes (fraction of body weight from Clewell 2001)

BW = 70.; % Body weight (kg)

VBLDC = 0.079; % Blood

VBONEC = 0.12; % Bone (Brown Human)

VBRNC = 0.02; % Brain 

VLIVC = 0.026; % Liver

VLUNGC = 0.008; % Lung 

VSTMC = 0.0015; % striatum (globus pallidus)

VCBMC = 0.08; % cerebellum

VOFBC = 0.00055; % olfactory bulb

VPITC = 0.0013; % pituitary (Dorman et al 2006)
42

rSANolf = 1.634; % monkey surface area nasal olfactory (cm2/kg) Schroeter 2008

rSANresp = 15.566; % monkey surface area nasal respiratory (cm2/kg) Schroeter 2008

SANOLF = 0.52; % surface area nasal olfactory (cm2/kg^.75) 

SANRESP = 9.81; % surface area nasal respiratory (cm2/kg^.75) % total SA=250 cm^2 for 70 kg

THNOSE = 375.; % thickness nasal tissue (um) Conolly (2000)

% lung and nasal deposition

% deposition fractions from MPPD for MnS04 - density = 2.95 g/cm^3,

% MMAD = 2.0, GSD = 1.5

FDEPNO = 0.376*0.05; % head deposition = 0.376, 5% olfactory airflow allocation

FDEPNR = 0.376*0.95; % head deposition = 0.376, 95% respiratory

FDEPLU = 0.25; 

% initials tissue concentrations (ug/kg) based on Sumino et al (1975)

COFB0 = 670.; % from Bonilla et al. (1982)

CSTM0 = 390.; % from Tracqui, Larsen, Aschner

CCBM0 = 330.; % from Bush, Aschner

CPIT0 = 460.; % from Thorladius-Ussing et al. (1988)

CLIV0 = 1200.; % from Sumino et al (1975)

CBONE0 = 74.;

CLUNG0 = 220.;

CART0 = 64.;

CVEN0 = 64.;

COTH0 = 90.;

CBILE0 = 0.;
43

% adjusted rate constants

KNPOBC=0.0035;

KDEPNRC=0.0012;

%liver

PLIV=1;

BMAXLIVC=1245;

KALIV=0.0011; 

KDLIV=0.00048; 

%lung

PLUNG=1; 

BMAXLUNGC=460;

KALUNG=0.01; 

KDLUNG=0.038; 

%bone

BMAXBONEC=78;

%body (others)

BMAXBODC=84;

%cerebellum

BMAXCBC=370;

KACB=0.001;

KDCB=0.003;

KINCBC=2.4;

KOUTCBC=0.08;
44

% globus pallidus

KAST=0.2;

KDST=0.02;

BMAXSTC=500;

KINSTC=8.0e-2;

KOUTSTC=4.0e-1;

KMI=0.16;

KSTLOSS=0;

KINMAX=4.6;

% pituitary

FPIT=0.99;

HP = 0.0;

BMAXPTC=620;

KAPT=0.3;

KDPT=0.036;

KINPTC=7.9e-2;

KOUTPTC=3.0e-1;

KMP=0.2;

KIPMAX=3.5;

KPTLOSS=0.;

%olfactory bulb

BMAXOBC=620;

KAOB=0.35;

KDOB=0.005;
45

KINOBC=3.0;

KOUTOBC=7.5;

% dietary controls

QBILE = 60; % bile flow rate (ml/h)

VBILE = 20.0; % gallblader volume 20 ml

FDIETUP = 0.06; % Diet absorption fraction

KBILEC = 0.051; % First order rate constant (/hr-1kg)

BDIET = 3.0; % mg/day

DDIET = 0.0;

INFAC = 0.05; % Dietary Intake Factors (kg/day/BW) (from NRC 2003)

% GI parameters

FENT = 0.005; 

KGI = 0.026;

KENT = 1.01*0.0022;

% exposure controls

DSTART = 100.; % acclimate for 100 days

DSTOP = 190.; % end exposure

DAYSON = 5.; % 5 days/week

TLEN = 8.; % 8 hours/days

TSTOP = 250.; % end simulation

KBINDUC = 2.5; 

KM = 0.027; 

N=3;
46

% ***************************************************************************

% m-file to simulate monkey inhalation exposure

% Simulate 65 exposure days 6 hours/day 5 days/week at 0 to 1.5 mg/m3 Mn on a 133 ppm diet

initials % reset parameters

monkey

DDIET = 133;

FDIETUP = 1.7e-3;

% simulate dose exposure and store concentration data

% inhalation 0.0 mg/m3 

DINH=0; start @NoCallback

y = _ctotliv; yy=_ctotlung; ycb=_ctotcb; 

yo=_ctotob; ys=_ctotst; yb=_ctotob; ybi = _cbile;

yp = _ctotpt; yc=_cart;

xt = _days;

% inhalation 0.06 mg/m3 

DINH=0.06; start @NoCallback

y06 = _ctotliv; yy06=_ctotlung; ycb06=_ctotcb; 

yo06=_ctotob; ys06=_ctotst; yb06=_ctotob; ybi06= _cbile;

yp06 = _ctotpt; yc06=_cart;

% inhalation 0.3 mg/m3 

DINH=0.3; start @NoCallback
47

y3 = _ctotliv; yy3=_ctotlung; ycb3=_ctotcb; 

yo3=_ctotob; ys3=_ctotst; yb3=_ctotob; ybi3=_cbile;

yp3 = _ctotpt; yc3=_cart;

% inhalation 1.5 mg/m3 

DINH=1.5; start @NoCallback

y15 = _ctotliv; yy15=_ctotlung; ycb15=_ctotcb; 

yo15=_ctotob; ys15=_ctotst; yb15=_ctotob; ybi15 = _cbile;

yp15 = _ctotpt; yc15=_cart;

% liver conc'

%plot(xt,y,x,y15,dv,mliv,'o', '.\aps\mk65.aps')

% lung conc'

%plot(xt,yy,x,yy15,dv,mlun,'o', '.\aps\mk65.aps')

% cerebellum conc'

%plot(xt,ycb,x,ycb15,dv,mcbm,'o', '.\aps\mk65.aps')

% olfactory bulb conc'

%plot(xt,yo,x,yo15,dv,mofb,'o', '.\aps\mk65.aps')

% striatum conc'

plot(xt,ys,x,ys15,dv,mgpd,'o' , '.\aps\mk65.aps')

% bile conc'

%plot(xt,ybi,x,ybi15,dv,mbile,'o', '.\aps\mk65.aps')
48

% pituitary conc'

%plot(xt,yp,x,yp15,dv,mpit,'o', '.\aps\mk65.aps')

% blood conc'

%plot(xt,yc,x,yc15,dv,mbld,'o', '.\aps\mk65.aps')
49

% **********************************************************************************

% m-file to run human inhalation exposure

% Simulate continuous 90 day exposure 8 hours/day 5 day/week for HUMAN

% simulate dose exposure and store concentration data

% inhaled concentration array mg/m3

v = [ 0.0001 0.001 0.01 0.02 0.03 0.04 0.06 0.08 0.1 0.2 0.3 0.5 0.7 1. 1.1 1.3 1.5 1.8 2 3 4 5 6 7 8 9 10];

[aa, bb] = size(v);

ddt = [0.0001 0.06 0.3 1.5];

ddd = [0.0001 0.01 0.1 0.5];

ddz = [0.0001 0.03 0.3 3];

disp('Running human simulations')

% reset parameters

initials

human

% exposure controls

DSTART = 100.; % acclimate for 100 days

DSTOP = 190.; % end exposure

DAYSON = 5.; % 5 days/week

TLEN = 8.; % 8 hours/day

TSTOP=290;
50

KBINDUC = 2.5; 

KM = 0.027; 

N=3;

BDIET=3;

FDIETUP = 0.06;

KBILEC = 0.051;

DINH=v(1); 

start @NoCallback

yhst = CTOTST;

yhf = ABRST/(VSTM*1000);

yhb = BMNST/(VSTM*1000);

yhcb = CTOTCB;

yhfcb = ABRCB/(VCBM*1000);

yhbcb = BMNCB/(VCBM*1000);

yhob = CTOTOB;

yhfob = ABROB/(VOFB*1000);

yhbob = BMNOB/(VOFB*1000);

yhpt = CTOTPT;

yhfpt = ABRPT/(VPIT*1000);

yhbpt = BMNPT/(VPIT*1000);

yhliv = CTOTLIV;

yhfliv = AVL/(VLIV*1000);

yhbliv = BMNLIV/(VLIV*1000);

yhlung = CTOTLUNG;

yhbone = CTOTBONE;

yhoth = CTOTOTH;
51

% run inhalation exposure loop

for i=2:bb

if (v(i) == 1)

 y15st = _ctotst;

 y15cb = _ctotcb;

 y15ob = _ctotob;

 y15pt = _ctotpt;

 y15liv = _ctotliv; 

 y15lung = _ctotlung;

 y15bone = _ctotbone; 

 y15oth = _ctototh;

end

DINH = v(i)

start @NoCallback

yhst(i)=CTOTST;

yhf(i) = ABRST/(VSTM*1000);

yhb(i) = BMNST/(VSTM*1000);

yhcb(i) = CTOTCB;

yhfcb(i) = ABRCB/(VCBM*1000);

yhbcb(i) = BMNCB/(VCBM*1000);

yhob(i) = CTOTOB;

yhfob(i) = ABROB/(VOFB*1000);

yhbob(i) = BMNOB/(VOFB*1000);

yhpt(i) = CTOTPT;

yhfpt(i) = ABRPT/(VPIT*1000);

yhbpt(i) = BMNPT/(VPIT*1000);

yhliv(i) = CTOTLIV;

yhfliv(i) = AVL/(VLIV*1000);

yhbliv(i) = BMNLIV/(VLIV*1000);
52

yhlung(i) = CTOTLUNG;

yhbone(i) = CTOTBONE;

yhoth(i) = CTOTOTH;

end

% plot figures

%semilogx(v, yhst,'g',v, ymst,'r', ddt, dgpd, 'or',v, yrst,'b', ...

% ddd, brain90(:,4), 'ob', ddz, zayed(:,5), 'oc','.\aps\rk65.aps')

% globus pallidus

semilogx(v, yhst, v, yhf, v, yhb)

plot(_days,y15st)

% cerebellum

%semilogx(v, yhcb, v, yhfcb, v, yhbcb)

%plot(_days,y15cb)

% olfactory bulb

%semilogx(v, yhob, v, yhfob, v, yhbob)

%plot(_days,y15ob)

% pituitary

%semilogx(v, yhpt, v, yhfpt, v, yhbpt)

%plot(_days,y15pt)

% liver

%semilogx(v, yhliv, v, yhfliv, v, yhbliv)

%plot(_days,y15liv)
53

%semilogx(v, yhlung)

%plot(_days,y15lung)

%semilogx(v, yhbone)

%plot(_days,y15bone)

% lung

%semilogx(v,yhoth)

%plot(_days,y15oth)